
#include <machine/asm.h>

#ifdef _ARM_ARCH_7

RCSID("$ : neon_mask.S,v 1.2 2020/12/18 06:14:23 matt Exp $")

/*
 * __neon_loading_qword_bitmask(size_t len);
 *  IN	r0 = length of mask in bits
 * OUT	q0 = mask
 */
ENTRY(__neon_leading_qword_bitmask)
	cmp		r0, #64		/* which dword is partial? */
#ifdef __ARMEL__
	sublt		r0, r0, #64	/* 1st dword needs MSBs cleared */
	subge		r1, r0, #128	/* 2nd dword needs MSBs cleared */
#else
	rsblt		r0, r0, #64	/* 1st dword needs LSBs cleared */
	rsbge		r1, r0, #128	/* 2nd dword needs LSBs cleared */
#endif
	movge		r0, #0		/* 1st dword needs to left alone */
	movlt		r1, #64		/* 2st dword needs to be cleared */
	vmov		d2, r0, r1	/* move dword shifts to SIMD */
	vmovl.u32	q1, d2		/* 2 U32 -> 2 U64 */
	vmvn.u64	q0, #0		/* create a mask */
	vshl.u64	q0, q0, q1	/* shift out unneeded bytes */
	RET
END(__neon_leading_qword_bitmask)

#endif	/* _ARM_ARCH_7 */
